
1. Overview

3. Handling of IMS SIP de-registration procedure when moving to other RAT
Change 1 – Correction to function ‘f_TC_13_4_3_14_IMS1’
	Function name
	f_TC_13_4_3_14_IMS1 ()

	Reason for change
	1.    IMS de-register needs to be handled in parallel to the handover procedure. 
2.    If UE has de-registered for IMS services during handover then early media removal should not be done.

3.    If UE has de-registered for IMS services during handover then the de-registration procedure at switch off should not be done.

	Summary of change
	1.    Added handling for SIP:DE-REGISTER message during handover.
2.    Early media removal will be done based on whether the UE has already de-registered for IMS services.
3.    Handling of SIP:DE-REGISTER at switch off is based on whether the UE has already de-registered for IMS services.

	TTCN module
	\13\ASRVCC_EUTRA_IMS.ttcn

	MCC160 Comment
	Accepted in principle and implemented under prose CR r5-154047


Before change

...
function f_TC_13_4_3_14_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;

    f_IMS_PTC_Init();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_UT_RequestIMSCall(MMI, px_IMS_CalleeUri); // @sic R5-134290 sic@

    /* Table 13.4.3.7.3.2-3: Parallel behaviour

       Steps 5-11 expected sequence defined in annex C.21 of TS 34.229-1 [35].

       NOTE: IMS MO speech call is in alerting phase.

    */

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); // @sic R5s140572 sic@

    /* TS 36.508 4.5A.10.3-1, parallel behaviour

     *  Steps 5 to 11 expected sequence defined in annex C.21

     */

    fl_IMS_MOCallSetup_AnnexC21_Steps5To11(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);

    f_IMS_SRVCC_RemoveEarlyDialogMO_AnnexC34(v_InviteRequestWithSdp); // @sic R5-144745 sic@

    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-142919 sic@

  }
...
After change

...
function f_TC_13_4_3_14_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;

    var boolean v_deRegistered;

    f_IMS_PTC_Init();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_UT_RequestIMSCall(MMI, px_IMS_CalleeUri); // @sic R5-134290 sic@

    /* Table 13.4.3.7.3.2-3: Parallel behaviour

       Steps 5-11 expected sequence defined in annex C.21 of TS 34.229-1 [35].

       NOTE: IMS MO speech call is in alerting phase.

    */

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); // @sic R5s140572 sic@

    /* TS 36.508 4.5A.10.3-1, parallel behaviour

     *  Steps 5 to 11 expected sequence defined in annex C.21

     */

    fl_IMS_MOCallSetup_AnnexC21_Steps5To11(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

    v_deRegistered := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN();
    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);

    if(not v_deRegistered) {
      f_IMS_SRVCC_RemoveEarlyDialogMO_AnnexC34(v_InviteRequestWithSdp); // @sic R5-144745 sic@

    }
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    if(not v_deRegistered) {
      f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-142919 sic@

    }
  }
... 
Change 2 – Correction to function ‘f_TC 13_4_3_14_UTRAN’
	Function name
	f_TC_13_4_3_14_UTRAN ()

	Reason for change
	Trigger should be sent from UTRAN PTC to IMS PTC to notify that handover is now complete and IMS PTC should stop waiting for SIP:DE-Register message. 

	Summary of change
	Added trigger to be sent to f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC() in IMS_PTC to stop waiting for IMS de-register message.

	TTCN module
	\13\ASRVCC_EUTRA_UTRAN.ttcn

	MCC160 Comment
	Accepted


Before change

...
  function f_TC_13_4_3_14_UTRAN() runs on UTRAN_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 36.108 Table 4.8.3-1 condition UTRA Speech

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 18" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5); // ToDo Cs domain

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); // @sic R5-144745, R5-150398 sic@

    v_TI_S.tiVal  := '000'B;

    v_TI_S.tiFlag := '0'B;

    //@siclog "Step 30" siclog@

    U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_Connect(v_TI_S)));

    v_TI_R := v_TI_S;

    v_TI_R.tiFlag := '1'B;

    //@siclog "Step 31" siclog@

    U_Dc.receive (car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_ConnAck(v_TI_R)));

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // @sic R5-144745 sic@

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }
...

After change

...
function f_TC_13_4_3_14_UTRAN() runs on UTRAN_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 36.108 Table 4.8.3-1 condition UTRA Speech

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 18" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5); // ToDo Cs domain

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); // @sic R5-144745, R5-150398 sic@

    v_TI_S.tiVal  := '000'B;

    v_TI_S.tiFlag := '0'B;

    //@siclog "Step 30" siclog@

    U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_Connect(v_TI_S)));

    v_TI_R := v_TI_S;

    v_TI_R.tiFlag := '1'B;

    //@siclog "Step 31" siclog@

    U_Dc.receive (car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_ConnAck(v_TI_R)));

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);  //Tell the IMS PTC to stop waiting for de-register message
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // @sic R5-144745 sic@

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }
... 
Change 3 – Correction to release with optional IMS deregistration
	Function name
	f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN ()
f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC ()

	Reason for change
	Status of whether the UE de-registered or not is required in f_TC_13_4_3_14_IMS1() to control early media removal procedure correctly. 

	Summary of change
	Added a Boolean return type to these functions with status of whether UE has de-registered from IMS services.

	TTCN module
	\Common\IMS\IMS_Procedures_Registration.ttcn

	MCC160 Comment
	Accepted in principle


Before change

...
function f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(IMS_IPCAN_CO_ORD_PORT p_Port,

                                                                boolean p_AllowRetransmission := false,

                                                                boolean p_WaitForDeRegistration := true) runs on IMS_PTC

  { /* @sic R5s140710 - MCC160 Implementation: parameter p_AllowRetransmission sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) SipUrl v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());

    var boolean v_WaitForDeRegistration := p_WaitForDeRegistration;

    var default v_RetransmissionHandler := null;

    alt { // @sic R5-141113: de-registration is not mandatory sic@

      [v_WaitForDeRegistration] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl))) -> value v_IMS_DATA_REQ

        {

          if (p_AllowRetransmission) {

            v_RetransmissionHandler := activate(a_IMS_RetransmissionHandler(v_IMS_DATA_REQ));  /* when re-transmissions are allowed they may colide with f_Delay in f_IMS_TCP_CloseHandling

                                                                                                  or with the confirmation in f_IMS_CloseTCP (if any); */

          }

          f_IMS_PTC_Deregistration(v_IMS_DATA_REQ);

          v_WaitForDeRegistration := false;

          repeat; /* @sic R5s140175 sic@ */

        }

      [] p_Port.receive(cmr_IMS_IPCAN_Trigger) // @sic R5-142919 sic@

        {

        }

    }

    deactivate(v_RetransmissionHandler);

  }

...

function f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(boolean p_WaitForDeRegistration := true) runs on IMS_PTC

  {

    var boolean v_AllowRetransmission := true;

    f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(OtherIPCAN, v_AllowRetransmission, p_WaitForDeRegistration);

  }
...

After change

...
function f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(IMS_IPCAN_CO_ORD_PORT p_Port,

                                                                boolean p_AllowRetransmission := false,

                                                                boolean p_WaitForDeRegistration := true) runs on IMS_PTC return boolean 

  { /* @sic R5s140710 - MCC160 Implementation: parameter p_AllowRetransmission sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) SipUrl v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());

    var boolean v_WaitForDeRegistration := p_WaitForDeRegistration;

    var default v_RetransmissionHandler := null;

    var boolean v_deRegistered := false;
    alt { // @sic R5-141113: de-registration is not mandatory sic@

      [v_WaitForDeRegistration] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl))) -> value v_IMS_DATA_REQ

        {

          if (p_AllowRetransmission) {

            v_RetransmissionHandler := activate(a_IMS_RetransmissionHandler(v_IMS_DATA_REQ));  /* when re-transmissions are allowed they may colide with f_Delay in f_IMS_TCP_CloseHandling

                                                                                                  or with the confirmation in f_IMS_CloseTCP (if any); */

          }

          f_IMS_PTC_Deregistration(v_IMS_DATA_REQ);

          v_WaitForDeRegistration := false;

          v_deRegistered:= true;
          repeat; /* @sic R5s140175 sic@ */

        }

      [] p_Port.receive(cmr_IMS_IPCAN_Trigger) // @sic R5-142919 sic@

        {

        }

    }

    deactivate(v_RetransmissionHandler);

    return v_deRegistered;
  }

...

function f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(boolean p_WaitForDeRegistration := true) runs on IMS_PTC return boolean 

  {

    var boolean v_AllowRetransmission := true;

    var boolean v_deRegistered; 

    v_deRegistered := f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(OtherIPCAN, v_AllowRetransmission, p_WaitForDeRegistration);

    return v_deRegistered;
  }
... 
